# Maintainer: Kevin Mihelich <kevin@archlinuxarm.org>
# Contributor: Kiril Zyapkov <kiril.zyapkov@gmail.com>

buildarch=20

_pkgname=gst-omx
pkgname=${_pkgname}-rpi
pkgver=1.2.0
pkgrel=2
pkgdesc="Gstreamer OpenMAX IL wrapper Plugin for Raspberry Pi"
arch=('armv6h' 'armv7h')
license=('LGPL')
url="http://gstreamer.freedesktop.org/"
depends=('gst-plugins-base-libs')
makedepends=('raspberrypi-firmware-tools' 'python2')
options=(!libtool)
source=(${url}/src/$_pkgname/$_pkgname-$pkgver.tar.xz)
md5sums=('d24e8c0153c35dfefee3e26b1c2c35f8')

LDFLAGS+=' -L/opt/vc/lib'
CPPFLAGS+=' -I/opt/vc/include -I/opt/vc/include/IL -I/opt/vc/include/interface/vcos/pthreads -I/opt/vc/include/interface/vmcs_host/linux'

build() {
  cd $srcdir/$_pkgname-$pkgver
  ./configure --prefix=/usr --sysconfdir=/etc --localstatedir=/var \
    --with-package-name="GStreamer OpenMAX IL wrapper Plugin (ArchlinuxARM/RPi)" \
    --with-package-origin="http://www.archlinuxarm.org/" \
    --with-omx-target=rpi
  
  make
}

package() {
  cd $_pkgname-$pkgver
  make DESTDIR="${pkgdir}" install
}
